
POPIS DIGITÁNÍHO NÁVRHU POMOCÍ VHDL 

 
PRAVIDLA PRO PSANÍ SYNTETIZOVATELNÉHO KÓDU 

 
 
1. Účel tohoto dokumentu 
Následující pravidla se týkají problematiky jak správně popisovat obvod ve VHDL, tak aby se návrhář vyhnul 
problémům při jeho syntéze. Zároveň v�ak do značné míry pokrývají i problematiku digitálního návrhu obecně. 
Správné psaní syntetizovatelného VHDL toti� není mo�né bez důkladného pochopení toho jak obvod bude 
implementován po syntéze. Ačkoli předpokládám, �e čtenář je ji� seznámen se základními dovednostmi 
číslicového návrháře, následující kapitolu by neměl pro jistotu přeskakovat. Dal�ím předpokladem pro 
pochopení tohoto dokumentu je alespoň základní znalost syntaxe jazyka VHDL. 
 

2. Obecné zásady digitálního návrhu 
 
Zásada č.1 Znát úroveň abstrakce 
Číslicové obvody se dnes navrhují metodou popisu v jazyce VHDL. Tento jazyk umo�ňuje popsat obvod na 
různých úrovních abstrakce. Toho se při návrhu často vyu�ívá. Například při psaní simulačních modelů, kde 
nás nezajímá konkrétní implementace, mů�eme vystačit s plně funkčním popisem do značné míry připomínající 
klasické programování. Tam je mo�né pou�ít celé spektrum konstrukcí jazyka VHDL.  
Je třeba si té� uvědomit, �e VHDL původně vznikl jako simulační jazyk, kde mělo být mo�né postupně 
přecházet od modelů čistě funkčních k modelům podrobněj�ím a� dospějeme k výslednému popisu struktury 
obvodu. Číslicová syntéza je vlastně procesem dodaným a� později za účelem automatizace přechodu mezi 
úrovněmi abstrakce. Proto je důle�ité vědět mezi jakými úrovněmi abstrakce syntéza přechází. 
V současné době umí syntéza automaticky převést obvod popsaný v jazyce VHDL na úrovni přesunů mezi 
registry (Register Transfer Level = RTL ) do úrovně hradlové - netlistu základních elementů dané technologie. 
Z toho plyne potřeba definovat určitou podmno�inu konstrukcí jazyka VHDL, které jsou podporovány syntézním 
procesem. Zároveň je nutné aby návrhář pochopil co znamená �navrhovat obvod na úrovni RTL�. 
Základní charakteristikou úrovně RTL je striktní vymezení toho, co jsou paměťové elementy (registry) a 
kombinační logika. Čas na úrovni RTL je diskrétní a je definován �tiky� hodinových signálů. 
Pokud návrhář dodr�í stanovená pravidla dané úrovně, vyhne se celé řadě problémů. Popisy na vy��í úrovni 
zpravidla vedou na nesyntetizovatelný kód. (Vy��í úrovně se někdy nazývají �behaviorální� - je to svým 
způsobem nevhodný název neboť i na úrovni RTL popisujeme chování obvodu - angl. chovat se  = to behave). 
Popsat obvod na ni��í jakési pseudohradlové úrovni je sice také mo�né, ale kromě zbytečné pracnosti a 
nepřehlednosti tyto popisy často vedou na neoptimální výsledky po syntéze. 
 
Zásada č.2 Chápat způsob implementace 
Jak je uvedeno v předchozí zásadě, popisujeme obvod na úrovni RTL. Proto je nutné aby návrhář my�lenkově 
chápal jak bude jím popsaný blok implementován - musí tedy být schopen odli�it od sebe registry a kombinační 
logiku. Při vlastním popisu je mo�né tyto věci smě�ovat (např. je mo�né v jednom procesu popsat část registru 
a kombinační logiky), návrhář v�ak v�dy musí být schopen nakreslit schema daného bloku (s oddělením 
registrů a kombinační logiky). Proto platí zásada - nakrestlit schema bloku před jeho popisem v jazyce VHDL.  
Neboť začátečníci - zejména s praxí programování - velice často mají problémy s tímto způsobem uva�ování, 
doporučuji jim, aby kombinační logiku a registry od sebe oddělovali i při vlastním VHDL popisu (viz doporučení 
dal�í kapitoly).  
 
Zásada č.3 Navrhovat synchronně jak je to jen mo�né 
Čtenář mů�e nabýt dojmu, �e pojem synchronní návrh se stal jistým zaklínadlem. Méně často je v�ak 
vysvětleno, co se za tímto pojmem vlastně skrývá. Proto zde uvádím tzv. zlatá pravidla synchronního návrhu. 
Tato pravidla jsou rozvedena v dal�í kapitole podrobněji. Je třeba ov�em uvést, �e za v�ech okolností není 

mo�né dodr�ovat tato pravidla, případů, kde obvod nelze navrhnout synchronně je v�ak výrazně méně ne� si 
začínající návrhář obvykle myslí. 
 
  

7 pravidel synchronního návrhu 
 
1.  Na hodinové vstupy v�ech prvků jsou přivedeny pouze hodinové signály. 
2.  Jsou pou�ity pouze hodinové signály jednofázové s pevným kmitočtem. 
3.  V�echny klopné obvody jsou hranově řízené (flip-flop ne latch). 
4.  Asynchronní přednastavení a nulování  mů�e být pou�ito pouze k 

nastavení počátečního stavu klopných obvodů, ne v průběhu jejich bě�né 
funkce. 

5.  Zpětné vazby v kombinačních obvodech se v návrhu nevyskytují. 
6.  Výstupy z dekodérů a přenosy čítačů nejsou pou�ity jako hodinové 

signály. 
7.  Asynchronní signály jsou synchronizovány pomocí klopného obvodu před 

jejich rozvedením do více funkčních jednotek . 
 
Pro věčné pochybovače je�tě uvedeme několik důvodů pro synchronní návrh.  

 
Proč synchronní návrh ? 

! Obvody navr�ené při dodr�ení pravidel synchronního návrhu jsou snadno testovatelné. Asynchronní obvody 
obsahující zpětné vazby způsobují značné problémy pro produkční testování. 

! Synchronní obvody jsou odolněj�í vůči �umu vznikajícím v důsledku přeslechů, odrazů, kombinačních 
hazardů, nestability zemních potenciálů, vlivům teplot,  ... 

! Funkčnost návrhu nezávisí na konkrétním rozmístění a propojení, tím je dána pouze dosa�ená rychlost 
funkce. 

! Při dodr�ení pravidel synchronního návrhu je k ověření postačující funkční simulace, simulace časová slou�í 
pouze k ověření dosa�eného hodinového kmitočtu. 

! U čistě synchronních obvodů s jedním hodinovým signálem v�dy simulace funkční musí souhlasit se 
simulací časovou. 

 
Vý�e uvedené neznamená, �e synchronní návrh nemá své nevýhody. Problémem na vět�ích čipech je mimo 
jiné samotný globální rozvod hodinového signálu, který by měl do ka�dého klopného obvodu dorazit ve stejný 
okam�ik, synchronní obvody také mívají vy��í spotřebu. Proto nejsou tato pravidla stanovena jako dogma. 
Ka�dý návrhář by v�ak měl důkladně uvá�it, je-li nezbytné aby některé z těchto pravidel poru�oval. Při 
respektování pravidel synchronního návrhu se vyvaruje mnoha problémů. 

3. Zásady a doporučení  
 
V následujících kapitolách uvedeme sadu pravidel, kterými se řídí popis obvodu v jazyce VHDL pro syntézu. 
Pravidla se děli do dvou skupin - na zásady a doporučení. Zásady jsou striktní pravidla, ze kterých nelze 
ustupovat (v�imněte si, �e skutečných zásad není mnoho). Doporučení jsou méně striktní, ale výjimky by měly 
být důkladně zvá�eny. 
 

3.1. Obecné zásady a doporučení 
 
Doporučení č. 1 - Pou�ívej standardní popis registrů hranově řízených (flip-flop) 
Jako sekvenční elementy v návrhu je doporučeno pou�ívat téměř výhradně registry hranově řízené (anglicky 
flip-flops). Příklad 1 ukazuje doporučený standardní popis, který po syntéze bezpečně vede na hranově řízený 
klopný obvod typu D tedy DFF. Jiné popisy vyu�ívající příkaz WAIT nebo paralelní přiřazení (mimo proces) se 
nebudou pou�ívat v rámci firemních projektů. (Tyto příkazy mají sice stejné chování, jejich podpora u jiných 
syntézátorů v�ak není zaručena. Mohli bychom mít zbytečné problémy s portabilitou kódu.) 
Nedoporučujeme inicializovat signály v deklaraci. Tato počáteční hodnota není podporována syntézou. Jediný 
způsob jak nastavit registru počáteční hodnotu je pou�ití asynchronního resetovacího signálu. 
Některé technologie - např. FPGA poskytují buňky DFF se signálem clock enable. Jedná se o multiplexer 
předřazený D klopnému obvodu. Syntézní programy se zlep�ují v rozpoznávání signálů pou�itelných jako clock 
enable, příklad č.2 ukazuje doporučený styl popisu, který by měl vést na obvod DFF se signálem CE.  
 
Příklad č.1 Standardní popis DFF s asynchronním a synchroním resetem 
 
SIGNAL q : std_logic:=’0’;  -- pocatecni hodnota je syntezou ignorovana ! 
 
SIGNAL q : std_logic; -- spravna deklarace 
 
clkp: PROCESS(clk,res_asy) 
BEGIN 
 IF res_asy=’1’ THEN q<=’0’; 
 -- asynchronni reset - pocatecni nastaveni 

 ELSIF clk’event AND clk=’1’ THEN 
  IF res_syn=’1’ THEN q<=’0’; -- synchronni reset (kombinacni logika na vstupu 
DFF) 
    ELSE q<=d; -- normalni funkce 

    END IF; 
END IF; 
END PROCESS clkp; 
 
Poznámka : Z hlediska návrhu pro snadnou testovatelnost se doporučuje vyvést u ka�dého 
klopného obvodu asynchronní resetovací signál. (Resetem zde rozumíme asynchronní přednastavení do nuly 
nebo do jednicky.) 
 
Příklad č.2 Registr DFFCE s asynchronním resetem a signálem clock enable (spec. pro FPGA) 
 
clkp: PROCESS(clk,res_asy) 
BEGIN 
 IF res_asy=’1’ THEN q<=’0’; 
 -- asynchronni reset - pocatecni nastaveni 

 ELSIF clk’event AND clk=’1’ THEN 
  IF ce=’1’ THEN q<=d; -- clock enable 

   END IF; 
 END IF; 
END PROCESS clkp; 
 
Zásada č.1 - Nepou�ívej hladinově řízené obvody (latch) 
Ačkoli jsou registry typu latch men�í ne� typ flip-flop jejich pou�ití přiná�í do návrhu dal�í parametr. Kromě 
hodinové frekvence zále�í i na střídě hodinového signálu resp. na délce aktivního pulsu hodinového signálu. 
Proto se hladinově řízené obvody při RTL návrhu nepou�ívají. Výjimky ov�em existují - jedná se o případy z 
návrhu velkých ASIC obvodů (např. velká registrová pole, FIFO, 2 fázové synchronní RAM). FPGA návrh by se 
měl bez obvodů typu latch zcela obejít. O případném pou�ití obvodů typu latch rozhoduje příslu�ný projektový 
vedoucí. 

Poznámka: Často jsou latche generovány jako důsledek �patného popisu kombinační logiky. Syntézní 
nástroj hlásí to jako varování při spu�tění funkce READ. Před vlastní funkční verifikací se doporučuje načíst 
soubory do syntézního programu a zkontrolovat přítomnost zmíněných �warnings�. 
 
Příklad č.3  Latch syntetizován v důsledku chybějící větve ELSE 
 
komblog: PROCESS(a,b) 
BEGIN 
  IF a=’1’ THEN 
     c<=b; 
  -- chybejici vetev  ELSE  znamena, ze c si pri a=’0’ pamatuje hodnotu ! 
   END IF; 
END PROCESS komblog; 
     
Příklad č.4 Syntetizován latch v důsledku chybějícího přiřazení výstupu z 
 
komblog: PROCESS(c) 
BEGIN 
  CASE c IS 
    WHEN ´0´ => z<=’0’; q<=’1’; 
    WHEN others  => q<=’0’; 
       -- vystup z neni definovan => pamatuje si hodnotu => implementovan latchem 
  END CASE 
END PROCESS komblog; 
 
Příklad č. 5 Syntetizovány latche v důsledku neúplného přiřazení std_logic_vectoru  
 
ps_toggle: PROCESS (curr_toggle, 
  reop,me,packet_err, 
  r_pid,curr_endp) 
  BEGIN 
    IF (reop = '1' AND me = '1' AND 
        packet_err = '0') THEN 
      IF (r_pid = DATA0_p) THEN 
        next_toggle(conv_integer(curr_endp)) <= '1'; 
  -- chybi definovat ostatni bity next_togle !!!! 
      ELSE 
        IF (r_pid = DATA1_p OR r_pid = SETUP_p) THEN 
          next_toggle(conv_integer(curr_endp)) <= '0'; 
  -- chybi definovat ostatni bity next_togle !!!! 
        ELSE 
          next_toggle <= curr_toggle; 
        END IF; 
      END IF; 
    ELSE 
      next_toggle <= curr_toggle; 
    END IF; 
  END PROCESS ps_toggle; 
 
Poznámka : Problémům z příkladů 4 a 5 se lze vyhnout buď důsledným definováním stavu v�ech signálů ve 
v�ech větvích podmínek nebo přiřazením �default� hodnoty na začátku procesu. 
 
Doporučení č.2 : Nevytvářej cykly mezi procesy 
Výsledkem cyklu mezi procesy mů�e být asynchronní obvod (implementován z hradel nebo pomocí hladinových 
obvodů) nebo zpomalení simulace. 
Návrhář se snadno vyhne těmto problémům pokud nebude pou�ívat nepřehledné vnořené IF-THEN - ELSE 
sekvence (příklad č.6) a nebude do jednoho procesu sdru�ovat signály mající odli�né vstupy (viz proces cl1 v 
příkladu č. 7). 

Příklad č. 6 Cyklus mezi procesy vedoucí na asynchronní chování 
 
--a je vstup 
--b, c interni signaly 
 
cl1: PROCESS (a,b) 
BEGIN 
  IF a=’1’ THEN  
     IF  b=’0’ THEN c<=’1’; 
     ELSE c<=’0’; 
     END IF; 
  ELSE 
     c<=’1’; 
  END IF; 
END PROCESS cl1; 
 
cl2: PROCESS(c) 
BEGIN 
  IF c=’1’ THEN 
    b<=’0’; -- c zavisi na a,b ale b take zavisi na c !!! 
   ELSE 
    b<=’1’; 
  END IF; 
END PROCESS cl2;  
 
Příklad č.7 Cyklus mezi procesy zpomalující simulaci 
 
cl1: PROCESS(a,c) 
BEGIN 
  IF a=’1’ THEN  
                b<=’0’; 
                IF c=’0’ THEN d<=’1’; 
                ELSE d<=’0’; 
                END IF; 
  ELSE 
            B<=’1’; 
            D<=’0’; 
  END IF; 
END PROCESS cl1; 
 
 
cl2: PROCESS(b) 
BEGIN 
  IF B=’1’ THEN  
      C<=’0’; 
  ELSE 
      C<=’1’; 
  END IF; 
END PROCESS cl2; 
 
-- b=f(a), c=f(b), d=f(a,b) 
-- process cl1 se vykonava vzdy dvakrat ! 
 
Zásada č.2 V�dy pou�ívej úplný a minimální citlivostní seznam (sensitivity list) u procesů 
Proces je v simulátoru proveden při změně některého signálu v jeho citlivostním seznamu (neuva�ujeme zde 
procesy se příkazy typu WAIT, které se v syntetizovatelném kódu nepou�ívají). 
V případě, �e proces popisuje kombinační logiku, v citlivostním seznamu musí být uvedeny v�echny signály 
na pravé straně přiřazovacích příkazů nebo řídící signály podmínek (IF, CASE), tedy v�echny vstupy 
kombinační logiky. 
Chybějící signály jsou zpravidla hlá�eny syntézátorem jako varování. Syntézátor si obvykle doká�e chybějící 
signály �domyslet� na to v�ak nelze spoléhat a je třeba chybu napravit. Varování tohoto typu se nesmějí 
vyskytovat v logovacím souboru syntézy. 

Příklad č.8 Kombinační logika 
 
komblog: PROCESS(a,b,c) 
BEGIN 
  IF c=’1’ THEN d<=a AND b; 
    ELSE d<=’1’; 
  END IF; 
END PROCESS komblog; 
 
V případě, �e proces popisuje sekvenční logiku (registry řízené hranou) obsahuje citlivostní seznam pouze 
hodinový signál a asynchronní reset - viz příklad č.1. Procesy obsahující v citlivostním seznamu mix hodinových 
signálů a výstupů kombinační logiky svědčí buď o asynchronním návrhu nebo o nezku�enosti návrháře.  
 
Poznámka : Začínajícím návrhářům se doporučuje striktně oddělovat kombinační a sekvenční procesy, dokud 
jazyk dostatečně neovládnou. 
 
Doporučení č. 3 Preferuj signály před proměnnými 
Proměnné nebývaly v minulosti v�dy podporovány při syntéze. Dnes se situace(MTI to umi, LHR). Pou�ití 
proměnných ov�em �etří paměť a zvy�uje rychlost simulace. Zku�ený návrhář proto mů�e proměnné pou�ít 
např. pro odstranění cyklické závislosti procesu sama se sebou (příklad č. 9). 
 
Příklad č.9 Převodník z grayova kódu 
 
-- gray to binary conversion 
gray2bin: PROCESS(QINT) 
VARIABLE Qbinvar : STD_LOGIC_VECTOR(width-1 downto 0); 
BEGIN 
   Qbinvar(width-1):=QINT(width-1); 
   FOR i IN width-2 downto 0 LOOP 
      Qbinvar(i):=Qbinvar(i+1) xor QINT(i); 
   END LOOP; 
   Qbin<=Qbinvar; 
END PROCESS gray2bin; 
 
-- pou�ití signálu místo proměnné Qbinvar je mo�né ov�em vede na zacyklení procesu sama na sebe 
-- u jednodu��ích simulátorů to znamená, �e proces je vykonáván width2 -krát místo width-krát. 
 

3.2. Pou�ití multiplexerů a třístavových budičů 
Doporučení č. 4 Preferuj CASE před vnořeným IF-THEN-ELSE 
Toto doporučení se opírá dnes ji� o poměrně historickou zku�enost se syntézními programy, které na základě 
vnořeného IF-THEN-ELSE syntetizovaly kaskádní víceúrovňový multiplexer. Při popisu s vyu�itím příkazu CASE 
je výsledkem stromová implementace multiplexeru, která je pochopitelně rychlej�í. 
Experimenty se současnou verzí syntézátoru Leonardo Spectrum tuto zku�enost sice nepotvrzuje, ov�em z 
důvodů portability kódu zachováváme toto doporučení. 
Jiným argumentem ve prospěch příkazu CASE je vět�í simulační rychlost na některých typech simulátorů.  

Obrázek 1 Kaskádní multiplexer někdy vznikající syntézou příkladu č.10 

d

c

b

a

sel="10" sel="01" sel="00"

e

 
 
Příklad č. 10 Vnořený IF THEN ELSE vedoucí údajně na kaskádní multiplexer 
pr1: PROCESS(a,b,c,d,sel) 
BEGIN 
  IF (sel="00") THEN 
    e<=a; 
  ELSIF (sel="01") THEN 
    e<=b; 
  ELSIF (sel="10") THEN 
    e<=c; 
  ELSE 
    e<=d; 
  END IF; 
END PROCESS pr1; 

Obrázek 2 Stromový (neprioritní) multiplexer vzniklý syntézou příkazu CASE (v�dy) 

sel(0)

sel(1)

a

b

c

d

Příklad č. 11 Funkčně ekvivalentní logika z příkladu 10 zapsána příkazem CASE 
 
pr1: PROCESS(a,b,c,d,sel) 
BEGIN 
 CASE sel IS 
  WHEN "00" => e<=a; 
  WHEN "01" => e<=b; 
  WHEN "10" => e<=c; 
  WHEN "11" => e<=d; 
  WHEN others => e<='-'; 
 END CASE; 
END PROCESS pr1; 
 
Multiplexer lze rovně� popsat pomocí podmíněného signálového přiřazení jako v příkladě č. 12. Jako nevýhoda 
tohoto zápisu je uváděn fakt, �e na některých typech simulátoru je pomalej�í. Návrhář mů�e uvá�it pou�ití 
tohoto zápisu jako alternativy klasického IF-THEN-ELSE. Výhodou je, �e odpadá psaní hlavičky procesu. 
Pro rozvětveněj�í podmínky se doporučuje pou�ívat procesu a příkazu CASE. 
 
Příklad č. 12 Multiplexer potřetí tentokrát jako podmíněné přiřazení 
 
e<=a WHEN sel=”00” ELSE 
      b WHEN sel=”01” ELSE 
      c WHEN sel=”10” ELSE 
      d; 
 
Doporučení č. 5 Třístavové budiče pou�ívej s rozvahou (speciálně pro Xilinx FPGA návrh) 
Třístavové budiče jsou nabízeny některými technologiemi  např. FPGA Xilinx. Jejich pou�ití je v zásadě dvojí - 
jako alternativa multiplexeru a k realizaci obousměrných sběrnic. Budiče jsou ideální ve sběrnicově 
orientovaných architekturách, kde potřebujeme realizovat spojení ka�dý s ka�dým pro vět�í počet bloků.  
Návrháři mů�e přijít výhodné pou�ívat třístavové budiče proto�e nekonzumují bě�nou logiku a zdá se, �e 
nevná�ejí zpo�dění jako rozvětvený multiplexer (to mů�e být často jen zdání neboť dlouhé sběrnice s velkou 
kapacitní zátě�í jsou pomalé). 
Zku�enosti ukazují, �e existuje trade-off mezi pou�itím třístavových sběrnic a multiplexerů (a dvoubodových 
spojů). Hlavní nevýhodou velkého pou�ití třístavových budičů je fakt, �e v FPGA konzumují �long lines�, kterých 
je samozřejmě limitovaný počet. Zejména při návrhu obsahujícím hodně vzájemně izolovaných třístvových 
sběrnic (ale třeba i jednu 32 a více-bitovou sběrnici) se potom mů�e stát, �e ačkoli FPGA není příli� zaplněno, 
obvod se nepropojí. (FPGA Virtex posouvají hranici opět dále, ov�em ani tam se nevyplatí problém ignorovat.) 
Optimální se jeví i u sběrnicových systémů sdru�ovat několik vysílačů přes multiplexer ke společnému 
třístavovému budiči (multiplexer 1-2 nebo 1-4 příli� zpo�dění nepřidá). 
 
Pro popis třístavového budiče je osvědčené pou�ít stručný zápis z příkladu č. 13. 
 
Příklad č.13 Třístavový budič 
 
bus<= data_out WHEN oe=’1’ ELSE (OTHERS=>’Z’); 
 

3.3. Pou�ití konečných automatů (FSM) 
 
 Zásada č. 3  Při popisu automatu pou�ívej tři procesy - dva kombinační procesy - pro přechodovou funkci 
automatu a pro výstupní funkci automatu a jeden sekvenční proces pro popis paměťových elementů.  
Některý z kombinačních procesů se mů�e redukovat na podmíněné nebo prosté signálové přiřazení - typicky 
jde o popis výstupní funkce automatu. 
Důvodem je jednak přehlednost popisu a jednak zaručená syntéza očekávané struktury. 
 
Doporučení č.6 Pou�ívej výčtový (enumerated) typ pro reprezentaci stavu  
Pou�ití výčtového typu jednak zpřehledňuje funkční simulace, jednak umo�ňuje měnit kódování automatu při 
syntéze - viz dokument o postupu při syntéze. 
 
Zásada č.4 Automaty separuj v samostatném bloku od jiné logiky 
Separace automatů do samostatného bloku umo�ňuje jejich separátní optimalizaci při syntéze - více viz 
doporučení pro dekompozici obvodu do bloků. 

Příklad č. 14 Popis automatu metodou 3 procesů 
 
TYPE machine2_state_type IS ( 
    LOCKED, 
    FREE 
  ); 
 
SIGNAL machine2_current_state, machine2_next_state : machine2_state_type ; 
 
BEGIN 
 
machine2_clocked : PROCESS (clk, res) 
BEGIN 
    IF (res = '1') THEN 
      machine2_current_state <= FREE; 
      -- Reset Values 
    ELSIF (clk'EVENT AND clk = '1') THEN 
      machine2_current_state <= machine2_next_state; 
      -- Default Assignment To Internals 
    END IF; 
  END PROCESS machine2_clocked; 
 
machine2_nextstate : PROCESS (FRAME, Hit,LOCK, L_lock,machine2_current_state) 
BEGIN 
    CASE machine2_current_state IS 
    WHEN LOCKED => 
      IF ((FRAME AND LOCK)='1') THEN 
        machine2_next_state <= FREE; 
      ELSE 
        machine2_next_state <= LOCKED; 
      END IF; 
    WHEN FREE => 
      IF (((NOT FRAME AND LOCK AND Hit AND (STATE_IDLE OR STATE_TURN_AR)) OR 
(L_lock AND Hit AND STATE_B_BUSY))='1') THEN 
        machine2_next_state <= LOCKED; 
      ELSE 
        machine2_next_state <= FREE; 
      END IF; 
    WHEN OTHERS => 
      machine2_next_state <= FREE; 
    END CASE; 
  END PROCESS machine2_nextstate; 
 
machine2_output : PROCESS (machine2_current_state) 
BEGIN 
    -- Default Assignment 
    -- Default Assignment To Internals 
    -- State Actions 
    CASE machine2_current_state IS 
    WHEN LOCKED => 
      STATE_LOCKED<='1'; 
      STATE_FREE<='0'; 
    WHEN FREE => 
      STATE_FREE<='1'; 
      STATE_LOCKED<='0'; 
    WHEN OTHERS => 
      NULL; 
    END CASE; 
 END PROCESS machine2_output; 

3.4. Dekompozice obvodu do bloků z hlediska syntézy 
Problém dekompozice obvodu do bloků je základní metodou pou�ívanou v průběhu standardní metodologie 
návrhu �shora-dolů�. Je obtí�né dát univerzální doporučení podle jakých hledisek obvod dekomponovat. 
Smyslem této kapitoly je stanovit pravidla na blokovou dekompozici z hlediska snadné a optimální syntézy. 
 
Základní algoritmus syntézy z úrovně RTL provádí převod abstraktního popisu kombinační a sekvenční logiky 
do konkrétní obvodové reperezentace - hradel a klopných obvodů dané technologie. S výjimkou některých 
speciálních bloků (jako např. násobiček) nemění polohu registrů vůči kombinační logice. Taková operace by 
vedla na obvod s odli�ným časovým rozvrhem operací - obvod RTL nenekvivalentní. Jedinou věcí, kterou 
syntéza optimalizuje je kombinační logika mezi registry (také ov�em eliminuje registry s trvalou konstantní 
hodnotou výstupu, provádí vyrovnávání zátě�e výstupů pomocí replikace logiky a pod.). Registry tvoří 
přirozenou bariéru pro optimalizaci kombinační logiky. 
Na hranice mezi bloky se vět�ina návrhářů dívá jako na virtuální hranice rozčleňující obvod na části my�lenkově 
pochopitelné části. Syntézátor ov�em implicitně chápe hranice mezi bloky jako dal�í umělé bariéry pro 
optimalizaci kombinační logiky. Z těchto dvou faktů plyne následující doporučení a zásada. 
 
Doporučení č. 7 Sna� se aby výstupy z bloků byly registrované 
Pokud chceme umo�nit syntéze optimalizovat kombinační logiku je nevhodné vkládat do ní umělou bariéru v 
podobě hranice bloku. Jinými slovy - výstupy z bloků by měly být výstupy z registrů - viz obrázek č. 3.  
Obvod dobře dekomponovaný dává potom lep�í výsledky při syntéze s přepínačem �preserve hierarchy� ne� 
�flatten�. Dal�ím důvodem registrovaných výstupů je usnadnění tvorby časových �constraints�. 
Je zřejmé, �e dodr�ovat toto doporučení striktně by značně omezovalo u�itečnou blokovou dekompozici. Proto 
moderní syntézní programy jako např. Leonardo Spectrum umo�ňují rozpou�tět hranice mezi bloky (disolve 
hiearchy). U vět�ích obvodů je v�ak nutné provádět optimalizaci na rozumně velkých blocích - při rozpu�tění 
hierarchie (flattening) celého obvodu dávají optimalizační algoritmy �patné výsledky nebo bě�í příli� dlouho.   
Proto zavádíme následující zásadu. 
 
Zásada č. 5 Bloky obsahující 1 tisíc a více ekvivalentních hradel musí mít registrované výstupy 
Dodr�ení této zásady nám umo�ní při syntéze rozpustit hierachii a� po učitou mez, kde je mo�no provádět 
optimalizaci na blokové úrovni. Leonardo Spectrum sice umo�ňuje spojovat bloky i mimo vlastní hierarchii, 
nelze na to v�ak spoléhat u jiných syntézních nástrojů. 
Při správném postupu dekompozice je vhodné té� provádět �time-budgeting� - vymezení dílu zpo�dění na daný 
blok. Bloky s registrovanými výstupy mají tento proces usnadněn. 
 
Doporučení č. 8 Sna� se dr�et kombinační logiku pohromadě v jednom bloku společně s příslu�nými 
registry 
Ve spojení s předchozími doporučeními se sna�íme naznačit, aby se návrhář sna�il vyvarovat samostatných 
bloků čistě kombinační logiky. Zejména by se neměly tyto bloky tzv. �glue logic� vyskytovat na nejvy��í úrovni 
hierarchie. Důvodem je opět omezení efektivity optimalizace a obtí�né časové rozpočtování (time-budgeting). 
 
Doporučení č. 9 Separuj časově kritickou logiku od časově nekritické 
Toto rozdělení umo�ní pou�tět odli�né optimalizační algoritmy na kritickou a nekritickou logiku. Typicky kritická 
logika je optimalizována na rychlost a nekritická na plochu. Volba typu algoritmu je mo�ná pouze na blokové 
úrovni proto je vhodné takto logiku separovat. 
 

3.5. Asynchronní logika 
Doporučení č. 10 Sna� se vyhnout asynchronní logice 
Asynchronní logika je daleko obtí�něji analyzovatelná a je mnohem obtí�něj�í navrhnout ji bezchybně a 
verifikovat ji. Implementace je obecně technologicky závislá z hlediska časových parametrů a to značně 
omezuje její portabilitu. 
 
Zásada č.6 O případné asynchronní implementaci rozhoduje vedoucí projektu a musí své rozhodnutí 
obhájit před týmem star�ích návrhářů. 
 
Doporučení č.11 Je-li asynchronní logika nezbytná,umísti jí do zvlá�tního bloku 
Separace asynchronní logiky umo�ní snadněj�í analýzu kódu. Asynchronní logika musí být analyzována 
mnohem pečlivěji z hlediska funkčnosti a časových parametrů. 

3.6. Aritmetické operátory 
Zásada č. 7 Pro aritmetické operátory pou�ívej knihovny IEEE.std_logic_arith nebo 
IEEE.std_logic_unsigned nebo IEEE.std_logic.signed. 
 
Výběr se řídí následujícími pravidly : 
a)  Pokud blok pracuje pouze s čísly bez znaménka pou�ívej knihovnu IEEE.std_logic_unsigned 
b)  Pokud blok pracuje pouze s čísly se znaménkem (v doplňkovém kódu) pou�ívej knihovnu 

IEEE.std_logic_signed 
c)  Pokud blok pracuje s čísly bez znaménka a se znaménkem pou�ij knihovnu IEEE.std_logic_arith 
 
Správný zápis aritmetické operace zále�í na pou�ité knihovně !!! 
V knihovně signed a unsigned  jsou definovány příslu�né aritmetické operace přímo pro typ std_logic_vector 
(Ten je chápán jako číslo se znaménkem nebo bez znaménka v závislosti na pou�ité knihovně). V knihovně 
arith je nutno explicitně přetypovat std_logic_vector na SIGNED nebo UNSIGNED. (Některé simulátory 
vy�adují i zpětné přetypování na std_logic_vector.) Výsledkem syntézy je příslu�ná jednotka (kombinační 
logika) pro čísla v daném formátu.  Příklad č. 5 ukazuje zápis sčítačky. 
 
Příklad č. 15 Popis sčítání  
 
a)  Pro knihovny std_logic_unsigned a std_logic_signed : 
 
SIGNAL a,b,c,d,z : std_logic_vector (...) 
 
IF ctl=´1´ THEN  
 z<= a + b; 
ELSE 
 z<= c + d; 
END IF; 
 
b)  Pro knihovnu std_logic_arith 
 
SIGNAL a,b,c,d,z : std_logic_vector (...) 
 
IF ctl=´1´ THEN  
 z<= unsigned(a) + unsigned(b); 
ELSE 
 z<= unsigned(c) + unsigned(d); 
END IF; 
 
Dal�í příklady uká�eme pro knihovnu std_logic_arith. 
 
Důle�ité je i správné přetypování na integer , zde při pou�ití SIGNED místo UNSIGNED dochází k odli�nému 
výsledku - viz příklad číslo 16. Vyu�ití typu integer  je nejčastěj�í v adresových dekodérech.  
Příklad č. 17 ukazuje popis obousměrného dvojkového čítače. V�imněte si, �e jednička je u inkrementace a 
dekrementace uvedena bez apostrofů - jde o operátor + 1 resp. -1, který je v knihovně �přetí�en� - definován pro 
typy integer, signed a unsigned.  
 
Příklad č. 16 Přetypování na integer a zpět 
 
SIGNAL a : std_logic_vector (3 downto 0); 
SIGNAL a_int : integer RANGE 0 to 15 
SIGNAL b : std_logic_vector (3 downto 0); 
 
BEGIN 
 
a_int<= conv_integer(unsigned(a)); 
 
b<= to_std_logic_vector(a_int,4); 
 
END; 

Příklad č. 17 Popis čítače 
 
IF up=’1’ THEN  
    nxt_cnt<= unsigned(cnt) + 1; 
ELSE 
    nxt_cnt<= unsigned(cnt) - 1; 
END IF; 
  
Doporučení č. 12 : Neseparuj aritmetické operátory do zvlá�tních bloků 
Charakteristickou vlastností aritmetických jednotek je mo�nost jejich sdílení pro více operací. Popis na příkladě 
č. 15 mů�e vést buď na syntézu dvou sčítaček s multiplexerem na výstupu nebo jedné sčítačky s multiplexery 
na vstupech. Syntézní program má tedy mo�nost optimalizace. V případě, �e sčitání je popsáno v separátním 
bloku, optimalizace je limitována hranicemi bloku a proto nedojde ke sdílení sčítaček. 
 
Závěrem je�tě několik poznámek : V�echny syntezátory předpokládají pou�ití řádové mří�ky celých čísel. V 
případě DSP aplikací se častěji pou�ívají jiné řádové mři�ky - např. tzv. zlomkový tvar. Návrhář proto musí mít 
elementární znalosti v oblasti číslicové aritmetiky (viz např. Pluháček A. : Projektování Logiky Počítačů). 
 
Několik otázek na které by si měl umět odpovědět ka�dý návrhář aritmetiky: 
 
1.  Lze pou�ít stejnou sčítačku/odčítačku pro čísla SIGNED a UNSIGNED ? 
2.  Lze pou�ít stejnou násobičku pro čísla SIGNED a UNSIGNED ? 
3.  Jak je třeba modifikovat sčítačku aby �la pou�ít jako odčítačka ? 
4.  Lze pou�ít stejnou násobičku pro čísla celá a ve zlomkovém formátu ? 

3.7. Dekompozice z hlediska rychlé syntézy a tvorba �constraints� 
V minulosti se doporučovalo pou�ívat k popisu relativně malé bloky k dosa�ení rozumné doby syntézy. Dnes 
vzhledem k růstu rychlosti syntézních algoritmů i pracovních stanic jsou hlavními hledisky dekompozice logická 
funkce a časové a plo�né po�adavky. Vhodná dekompozice společně s constraints má vět�í vliv na dobu 
syntézy ne� vlastní velikost bloku. Důle�ité je separovat logiku kritickou na rychlost od logiky kritické na plochu. 
To umo�ňuje pou�ívat odli�né optimalizace pro ka�dý blok a výrazně sni�uje dobu syntézy.  
Tvorba vhodných �constraints� je relativně komplexní problém, který není mo�né postihnout v celé �íři v rámci 
tohoto dokumentu. Důle�ité je vyhnout se nerealistickým po�adavkům - tzv. �overconstraining�. Pokud 
výsledkem běhu programu bez �constraints� je obvod o frekvenci 50 MHz je realistické očekávat, �e při 
vhodných �constraints� lze dosáhnout 60 - 70 MHz ov�em ji� velmi nepravděpodobně 100 MHz. (Pokud je cílem 
dosáhnout 100 MHz, je primárně nutné změnit RTL návrh s vyu�itím pipeliningu a pod.) 
Metodika tvorby constraints u vět�ích obvodů se nazývá časové rozpočtování (�time budgeting�). V ranných 
stádiích návrhu je vhodné rozpočíst zpo�dění jednotlivých bloků a na tomto základě vytvořit vhodné constraints. 
Stejné constraints potom slou�í i při verifikaci s vyu�itím statické časové analýzy. 
 

3.8. Fale�né a vícecyklové cesty 
Doporučení č.13 Vyhýbej se pou�ití vícecyklových cest 
Vícecyklovou cestou (Multicycle path, point to point exception) rozumíme cestu v kombinační logice mezi 
zdrojovým a cílovým registrem, kterou se signál mů�e �ířit po dobu více ne� jednoho hodinového taktu ne� je 
zapsán do cílového registru. 
Pou�ití vícecyklových cest s sebou nese problémy při tvorbě constraints, statické časové analýze a také při 
pochopení funkce návrhu. Proto je vhodné se vícecyklových cest vyvarovat. 
 
Doporučení č. 14 Pokud jsou pou�ity vícecyklové cesty, umísti je do jednoho modulu a dobře je 
zdokumentuj. Modulem rozumíme blok na vy��í úrovni hierarchie. U návrhu obsahující jak vícecyklové, tak 
jednocyklové cesty je vhodné tyto cesty od sebe oddělit. 
 
Doporučení č. 15  Vyhýbej se vzniku fale�ných cest 
Fale�ná cesta (False path) je cesta v kombinační logice nesplňující časový limit, která v�ak při reálné fukci 
obvodu nikdy není aktivována. Tyto cesty vznikají často ve sběrnicově orientovaných systémech a tam, kde 
mezi dvěma registry existuje více ne� jedna mo�ná cesta (typicky zpětné vazby v ALU). Fale�né cesty je nutné 
specifikovat do constraints neboť vedou na dlouhé časy při optimalizaci a statická časová analýza jinak dává 
�patné údaje. Návrh obsahující více fale�ných cest je nutno verifikovat časovou simulací a proto je vhodné se 
fale�ným cestám vyhýbat. 

3.9. Návrh obsahující paměti 
Paměti pou�íváme v návrhu FPGA a ASIC velice často. Ať u� k realizaci registrových polí nebo jako 
vyrovnávací a jiné paměti. Paměti ov�em představují problém při snaze o přenositelnost návrhu. Důvodem je 
fakt, �e typ paměti je značně technologicky závislý. Typicky jsou podporovány buď tzv. synchronní nebo 
asynchronní RAM s různým rozhraním.  
 
Nejprve provedeme malý úvod do terminologie, která je často velmi zmatená. Připomínám, �e se budeme 
zabývat pouze STATICKÝMI pamětmi RAM, které jsou integrovány na čipech s logikou. Integrace DRAM 
představuje stále je�tě značný technologický oří�ek.  
 

3.9.1. Přehled typů RAM 
1.  Asynchronní RAM  
Asynchronní RAM je mo�né z hlediska funkce pova�ovat za pole hladinově řizených klopných obvodů LATCH, 
se selektivním zápisem a čtením. 
 
Obrázek č. 3 Funkční pohled na asynchronní RAM 4 x 1  

DEC

D

LD

Q

&

D

LD

Q

&

D

LD

Q

&

D

LD

Q

&

Addr_wr

WE

DIN

RD

DOUT

Addr_rd  
 
Pro čtení se chová jako kombinační logika (signál RD a třístavový budič mů�e umo�nit spojení DOUT a DIN a 
společnou adresu pro čtení a zápis). Pro zápis je důle�ité vygenerovat příslu�ný zápisový puls WE a dodr�et 
stabilní signály Addr_wr a DIN. Výhodou asynchronní RAM je malá plocha, nevýhodou je obtí�něj�í generování 
řídicích signálů. 
 
2.  Synchronní RAM (té� RAM se synchronním zápisem) 
Synchronní RAM je mo�né z hlediska funkce pova�ovat za pole hladinově řízených klopných obvodů DFF, 
se selektivním zápisem a čtením.  
V praxi se z důvodu úspory plochy realizuje jako Master / Slave, kde opět mů�eme vyu�ít vý�e zmíněnou 
asynchronní RAM. 
Na obrázku č. 4 je ukázka principiálního zapojení synchronní RAM s pou�itím asynchronní RAM. V�imněme si, 
�e nyní přibyl hodinový signál CLK. Z hlediska čtení se paměť opět chová jako kombinační logika.  

Obrázek č. 4 Realizace synchronní RAM s pou�itím asynchronní RAM 

3. Vícecyklové paměti 
Potřeba zajistit prediktabilní časování vede výrobce FPGA a ASIC k umísťování dal�ích registrů na adresové a 
datové signály. Pro návrháře to znamená, �e pro čtení i zápis je nutno často počítat s dodatečnou latencí. 
Obrázek č. 5 ukazuje příklad takové vícecyklové paměti. Čerchované čáry naznačují mo�ná umístění 
dodatečných registrů.  
 
Obrázek č. 5 Vícecyklové paměti 

Jednotliví výrobci FPGA označují jako paměť nejrůzněj�í kombinace dříve uvedených typů. Např. Xilinx řada 
4000 podporovala asynchronní distribuovanou RAM, řada 4000E synchronní a asynchronní distr. RAM, 
Spartany a nověj�í pouze synchronní distribuovanou RAM (ná� typ č. 2). Tento typ je označován jako Select 
RAM. Kromě toho řada Virtex přiná�í tzv. Block RAM, která má ov�em registrovaný výstup a kde adresy musí 
být generovány z registrů (tyto registry jsou implementovány na poli v CLB). Co označuje Xilinx jako Block RAM 
je ukázáno na obrázku č. 5.  
Konkurenční výrobce Altera pou�ívá pouze synchronní blokové paměti RAM v tzv. EAB (Embeded Array Block). 
Altera jako RAM označuje dokonce i registry na adresních vstupech, registr na výstupu naopak být nemusí. 
 
Z vý�e uvedeného je zřejmé, �e navrhnout obvod nebo core přenositelné mezi různými FPGA a ASIC je značně 
obtí�né. 
 

3.9.2. Paměti z hlediska syntézy 
Tradiční cesta syntézy pou�ívá pro RAM technologicky specifické RAM generátory. Syntézní nástroje potom s 
RAM zacházejí jako s black boxem. To ov�em znamená, �e RAM je bariérou statické časové analýzy. 
Syntezátor Leonardo Spectrum je schopen FPGA RAM rozpoznat přímo z RTL popisu.  
 
Pro snadnou přenositelnost návrhu mezi technologiemi i syntézními programy je vhodné dodr�ovat následující 
doporučení. 
 
Doporučení č. 16 Odděl řídící logiku od paměti do samostatného modulu 
Obrázek č. 6 ukazuje správnou dekompozici bloku s pamětí. Řídící logika by v�dy měla být umístěna v 
separátním bloku od vlastního popisu paměti. To umo�ňuje obě cesty syntézy RAM. Blok paměťové matice 
mů�e být chápán jako black box (syntéza se Synopsys, ASIC syntéza s Leonardem) nebo nahrazena RTL 
popisem a inferována z kódu (FPGA syntéza s Leonardem). 
 
Obrázek č. 6 Dekompozice bloku s pamětí 

Doporučení č. 17 Pou�ívej synchronní paměti RAM  
Ve spojení s předchozím doporučením to znamená, �e řídící logika by v�dy měla předpokládat synchronní 
paměť. V případě, �e daná technologie nepodporuje synchronní RAM přímo, mů�e být tato vytvořena s pou�itím 
glue logiky a asynchronní RAM. Tento příklad je ilustrován na obrázku č. 7. Standardní popis vedoucí na 
inferenci synchronní Select RAM pro Xilinx je uveden v příkladě č. 18.  
 
Doporučení č. 18 Sna� se, aby adresa byla generována z registru a/nebo aby data byla čtena přímo do 
registru 
Toto doporučení ve svém důsledku znamená, latenci 1 a� 2 takty pro čtení. Paměť splňující plně tento 
po�adavek je v�ak bez problémů přenositelná mezi jednotlivými typy FPGA a ASIC obvodů i syntézního 
software.  

Příklad č. 18 Standardní popis synchronní jednoportové Select RAM 
 
ENTITY spram IS 
GENERIC(widthA : integer := 4; 
            widthD : integer := 8); 
PORT( 
     addr  :  IN std_logic_vector(widthA-1 downto 0); 
     din   :  IN std_logic_vector(widthD-1 downto 0); 
     ce_wr :  IN std_logic; 
     wr_clk:  IN std_logic; 
     dout  :  OUT std_logic_vector(widthD-1 downto 0) 
     ); 
END spram; 
 
ARCHITECTURE rtl OF spram IS 
 
TYPE t_memory IS array (0 to 2**widthA-1) OF std_logic_vector(widthD-1 downto 0); 
 
SIGNAL memory : t_memory; 
 
BEGIN 
 
write_proc: PROCESS(WR_CLK) 
-- note : there is no reset in memory write description ! 
BEGIN 
 IF wr_clk'event AND wr_clk='1' THEN 
  IF CE_WR='1' THEN memory(conv_integer(unsigned(addr)))<=din; 
  END IF; 
 END IF; 
END PROCESS write_proc; 
 
read_proc: PROCESS(addr,memory) 
-- note there is no clock for read, memory is combinatorial for read ! 
BEGIN 
  dout <= memory(conv_integer(unsigned(addr))); 
END PROCESS read_proc; 
 
END rtl; 
 
Obrázek č. 7 Pou�ití asynchronní RAM 

Poznámka : Zvlá�tní pozornost je třeba věnovat RAM v případě tvorbě constraints pro statickou časovou 
analýzu. RAM často znamená hranici statické časové analýzy. Pro zápis zde obvykle nedochází k problémům, 
neboť synchronní paměť RAM je ekvivalentní poli hranově řízených D klopných obvodů. Při čtení se ov�em 
RAM chová jako kombinační logika. Pokud RAM přeru�uje cesty statické časové analýzy je mo�né, �e nemusí 
proběhnout korenktně zvlá�tě v případě, kdy adresa paměti je generována kombinační logikou. Proto je vhodné 
generovat adresu do paměti s pou�itím DFF. Jinak je nutné constraints specifikovat zvlá�tě pro zpo�dění do 
RAM a zpo�dění z RAM.  

4. Závěrem 
Tento dokument představuje základní návod jak popisovat obvod v jazyce VHDL, tak aby tento popis byl 
snadno syntetizovatelný a navíc přenositelný na různé technologie. Autor doufá, �e se mu podařilo začínajícího 
návrháře neznechutit seznamem omezení. Pokud text přesto tímto dojmem působí, autor zdůrazňuje, �e drtivá 
vět�ina omezení vychází často z tě�ce získaných negativních zku�eností z reálných projektů. Neboť v�echno se 
stále vyvíjí mohou některá omezení být časem změněna. Autor proto uvítá jakékoli připomínky k tomuto textu. 
 
 

5. Revize dokumentu 
 
Číslo rev. Autor Datum Popis 
1.0 MBE 25.5. Úvodní revize pro první kolo interních �kolení 
2.0 MBE   23.11. Kapitola o aritmetice přeformulována, zásada č. 7 změněna. 

Sní�ena velikost bloku v zásadě č. 5 (zku�enosti s ASIC syntézou) 
Kapitola o RAM doplňena o výklad a doporučení č. 18 

    
    
    
 


